
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2022-09_D22wk49 related to the title of this CR.
Contact:	Shaun Harry
	shaun.harry@keysight.com 	

2 [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc123819622]Corrections required
2. [bookmark: _Toc106975634][bookmark: _Toc109680267][bookmark: _Toc122623878][bookmark: _Toc123819623][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665]Change 1
	Function name
	f_TC_7_1_NR5GC_IMS1

	Reason for change
	When the testcase excution does not go down optional path 14a/b, ie. no call is setup, the testcase will fail at postamble where TTCN try to release the call

	Summary of change
	Added check on the excution path, and call postamble with appriciated procedure accordingly. 

	TTCN module
	IMS_NR5GC_CallControlTestcases.ttcn

	MCC160 Comment
	



Before Change
	  function f_TC_7_1_NR5GC_IMS1() runs on IMS_PTC
  { // MTSI MO Voice Call / 503 Service Unavailable / 5GS
    var template (present) SipUrl v_CalleeUri := f_IMS_INVITE_CalleeUriRX();    // as used per default in the INVITE and in this test case for the ACK as the UE has not gotten any callee's contact URI
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var INVITE_Request v_InviteRequest;
    var template (value) MessageHeader v_MessageHeader503;
    var template (present) MessageHeader v_MessageHeaderACK;
    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;
    var integer v_RetryAfter := 20;   // @sic R5-210882: 60s -> 20s sic@
    var float v_RetryAfterMin := f_IMS_SetTimerToleranceMin(int2float(v_RetryAfter));
    var boolean v_InviteWithPreconditions;
    
    timer t_Timer;

    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED);
    
    f_IMS_TestBody_Set(true);

    // @siclog "Step 1" siclog@
    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall);

    // @siclog "Step 2 - 7" siclog@
    v_InviteWithPreconditions := fl_IMS_MOCallSetup_CheckInviteWithPreconditions();

    if (v_InviteWithPreconditions) {  // @sic R5-213521 sic@
      // @siclog "Step 8a" siclog@              INVITE
      v_InviteRequestWithSdp := f_IMS_MOCallSetup_NR5GC_A_4_1a_Step1(v_CalleeUri);               // @sic R5-225925: A_4_1 -> A_4_1a sic@
    } else {
      // @siclog "Step 8b" siclog@              INVITE
      v_InviteRequestWithSdp := f_IMS_MOCallSetup_NR5GC_A_4_2_Step1(v_CalleeUri);
    }
    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    // @siclog "Step 9" siclog@                 503 Service Unavailable
    v_MessageHeader503 := f_IMS_CommonResponse_503_MessageHeaderTX(v_InviteRequest.msgHeader, cs_RetryAfter(v_RetryAfter));
    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_InviteRequestWithSdp.RoutingInfo);
    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine503, v_MessageHeader503)));
    
    // @siclog "Step 10" siclog@                ACK
    v_MessageHeaderACK := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine503);
    f_IMS_ReceiveRequest(car_IMS_Ack_Request(cr_ACK_Request(v_CalleeUri, v_MessageHeaderACK)));
    
    f_IMS_PTC_ImsInfo_DialogRelease();

    // @siclog "Step 11 - 13" siclog@           Check: Does the UE (not) transmit INVITE request message
    f_Delay(v_RetryAfterMin);
    
    t_Timer.start(30.0);
    alt {                     // @sic R5-216210 sic@
      [] IMS_Server.check(receive(car_IMS_Invite_Request)) {
        t_Timer.stop;
        if (v_InviteWithPreconditions) {  // @sic R5-213521 sic@
          // @siclog "Step 14a1" siclog@            INVITE
          v_InviteRequestWithSdp := f_IMS_MOCallSetup_NR5GC_A_4_1a_Step1(v_CalleeUri);          // @sic R5-225925: A_4_1 -> A_4_1a sic@
          f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 14a1");
          // @siclog "Step 14a2-14a15" siclog@      Steps 2 to 12 of A.4.1
          f_IMS_MOCallSetup_NR5GC_A_4_1a_Step2_12(v_InviteRequestWithSdp);                      // @sic R5-225925: A_4_1 -> A_4_1a sic@
        } else {
          // @siclog "Step 14b1" siclog@            INVITE
          v_InviteRequestWithSdp := f_IMS_MOCallSetup_NR5GC_A_4_2_Step1(v_CalleeUri);
          f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 14b1");
          // @siclog "Step 14b2-14b8" siclog@       Steps 2 to 8 of subclause A.4.2
          f_IMS_MOCallSetup_NR5GC_A_4_2_Step2_8(v_InviteRequestWithSdp);
        }
      }
      [] t_Timer.timeout {}
    }

    f_IMS_TestBody_Set(false);

    // Call release and postamble:
    f_IMS_CC_PostambleWithCallReleaseMT(IPCAN_MO_SpeechCall);
  }



After Change
	  function f_TC_7_1_NR5GC_IMS1() runs on IMS_PTC
  { // MTSI MO Voice Call / 503 Service Unavailable / 5GS
    var template (present) SipUrl v_CalleeUri := f_IMS_INVITE_CalleeUriRX();    // as used per default in the INVITE and in this test case for the ACK as the UE has not gotten any callee's contact URI
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var INVITE_Request v_InviteRequest;
    var template (value) MessageHeader v_MessageHeader503;
    var template (present) MessageHeader v_MessageHeaderACK;
    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;
    var integer v_RetryAfter := 20;   // @sic R5-210882: 60s -> 20s sic@
    var float v_RetryAfterMin := f_IMS_SetTimerToleranceMin(int2float(v_RetryAfter));
    var boolean v_InviteWithPreconditions;
    var boolean v_needCallRelease;
    
    timer t_Timer;

    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED);
    
    f_IMS_TestBody_Set(true);

    // @siclog "Step 1" siclog@
    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall);

    // @siclog "Step 2 - 7" siclog@
    v_InviteWithPreconditions := fl_IMS_MOCallSetup_CheckInviteWithPreconditions();

    if (v_InviteWithPreconditions) {  // @sic R5-213521 sic@
      // @siclog "Step 8a" siclog@              INVITE
      v_InviteRequestWithSdp := f_IMS_MOCallSetup_NR5GC_A_4_1a_Step1(v_CalleeUri);               // @sic R5-225925: A_4_1 -> A_4_1a sic@
    } else {
      // @siclog "Step 8b" siclog@              INVITE
      v_InviteRequestWithSdp := f_IMS_MOCallSetup_NR5GC_A_4_2_Step1(v_CalleeUri);
    }
    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    // @siclog "Step 9" siclog@                 503 Service Unavailable
    v_MessageHeader503 := f_IMS_CommonResponse_503_MessageHeaderTX(v_InviteRequest.msgHeader, cs_RetryAfter(v_RetryAfter));
    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_InviteRequestWithSdp.RoutingInfo);
    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine503, v_MessageHeader503)));
    
    // @siclog "Step 10" siclog@                ACK
    v_MessageHeaderACK := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine503);
    f_IMS_ReceiveRequest(car_IMS_Ack_Request(cr_ACK_Request(v_CalleeUri, v_MessageHeaderACK)));
    
    f_IMS_PTC_ImsInfo_DialogRelease();

    // @siclog "Step 11 - 13" siclog@           Check: Does the UE (not) transmit INVITE request message
    f_Delay(v_RetryAfterMin);
    
    t_Timer.start(30.0);
    v_needCallRelease := false;
    alt {                     // @sic R5-216210 sic@
      [] IMS_Server.check(receive(car_IMS_Invite_Request)) {
        t_Timer.stop;
        if (v_InviteWithPreconditions) {  // @sic R5-213521 sic@
          // @siclog "Step 14a1" siclog@            INVITE
          v_InviteRequestWithSdp := f_IMS_MOCallSetup_NR5GC_A_4_1a_Step1(v_CalleeUri);          // @sic R5-225925: A_4_1 -> A_4_1a sic@
          f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 14a1");
          // @siclog "Step 14a2-14a15" siclog@      Steps 2 to 12 of A.4.1
          f_IMS_MOCallSetup_NR5GC_A_4_1a_Step2_12(v_InviteRequestWithSdp);                      // @sic R5-225925: A_4_1 -> A_4_1a sic@
        } else {
          // @siclog "Step 14b1" siclog@            INVITE
          v_InviteRequestWithSdp := f_IMS_MOCallSetup_NR5GC_A_4_2_Step1(v_CalleeUri);
          f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 14b1");
          // @siclog "Step 14b2-14b8" siclog@       Steps 2 to 8 of subclause A.4.2
          f_IMS_MOCallSetup_NR5GC_A_4_2_Step2_8(v_InviteRequestWithSdp);
        }
        v_needCallRelease := true;
      }
      [] t_Timer.timeout {
        f_IMS_CC_AbortCall();
      }
    }

    f_IMS_TestBody_Set(false);

    // Call release and postamble:
    if(v_needCallRelease) {
      f_IMS_CC_PostambleWithCallReleaseMT(IPCAN_MO_SpeechCall);
    } else {
      f_IMS_CC_Postamble(IPCAN_MO_IMS_Signalling);
    }
  }






